Model-Checking

CS511

Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model

Assertions

Non-Progress Cycles

End States

Model-Checking: Stating Properties Outside the Mode

Program Correctness

Main approaches to demonstrating that a program does what it's supposed to do:

- 1. Testing
- 2. Deductive verification
- 3. Model-checking

Testing

- Fast and simple way to detect errors
- Can never be sure there are no defects (cannot cover all the cases) maybe tests weren't comprehensive enough
 Testing shows the presence, not the absence of bugs¹

Testing Concurrent Programs

- More difficult since we would like to test all interleavings
- ▶ But since interleaving is controlled by OS scheduler, user cannot arrange arbitrary interleavings
- Consequence: very few of possible interleavings are tested

¹Dijkstra (1969) J.N. Buxton and B. Randell, eds, Software Engineering Techniques, April 1970, p. 16. Report on a conference sponsored by the NATO Science Committee, Rome, Italy, 27–31 October 1969.

Proving Programs Correct

- ► Holy Grail of computer science
- Using special specification language, describe
 - 1. State of program's variables
 - 2. How each programming language statement uses variables
- Specification language is mixture of mathematics & programming language

How to Prove a Program Correct

Hoare Triples

$$\llbracket A \rrbracket P \llbracket B \rrbracket$$

- ► P program
- ► A precondition
- ▶ B postcondition
- ► A and B are predicate logic formulae over an extended first-order language

Example

```
1  y:=1;
2  z:=0;
3  while (z!= x) {
4   z:=z+1;
5   y:=y*z
6 }
```

Assertion

- ▶ Using hoare triples: $[x > 0] P[y = z! \land z = x]$
- ▶ In prose: Under any state σ such that x > 0, if P terminates in a state ρ , then ρ satisfies $y = z! \land z = x$

Provable Assertion

▶ Prove $[x > 0]P[y = z! \land z = x]$ in some deductive proof system

Sample Deductive Proof System for Partial Correctness

$$\frac{\llbracket A \rrbracket C_1 \llbracket B \rrbracket \quad \llbracket B \rrbracket C_2 \llbracket C \rrbracket}{\llbracket A \rrbracket C_1; C_2 \llbracket C \rrbracket} \text{(Composition)}$$

$$\frac{\llbracket A \setminus B \rrbracket C_1; C_2 \llbracket C \rrbracket}{\llbracket A \setminus B \rrbracket C_2 \llbracket A \rrbracket} \text{(Assignment)}$$

$$\frac{\llbracket A \wedge B \rrbracket C_1 \llbracket D \rrbracket \quad \llbracket A \wedge \neg B \rrbracket C_2 \llbracket D \rrbracket}{\llbracket A \rrbracket \text{if } B \text{ then } \{C_1\} \text{ else } \{C_2\} \llbracket D \rrbracket} \text{(Conditional)}$$

$$\frac{A \to A' \quad \llbracket A' \rrbracket P \llbracket B' \rrbracket \quad B' \to B}{\llbracket A \rrbracket P \llbracket B \rrbracket} \text{(Implication)}$$

$$\frac{\llbracket A \wedge B \rrbracket C \llbracket A \rrbracket}{\llbracket A \rrbracket \text{while } B \{C\} \llbracket A \wedge \neg B \rrbracket} \text{(While-Partial)}$$

Example of Partial Correctness Proof

$$[[x > 0]] y := 1; z := 0; \text{ while } (z! = x) \{z := z + 1; y := y * z\} [[y = z! \land z = x]]$$

$$\frac{ [\![y*(z+1)=(z+1)!]\!]z:=z+1[\![y*z=z!]\!] }{ [\![y=z! \land z \neq x]\!]z:=z+1[\![y*z=z!]\!] }$$

$$\frac{ [\![y=z! \land z \neq x]\!]z:=z+1[\![y*z=z!]\!] }{ [\![y=z! \land z \neq x]\!]z:=z+1;y:=y*z[\![y=z!]\!] }$$

$$[\![y=z!]\!]Q[\![y=z! \land z=x]\!]$$

$$Q = \text{while}(z!=x) \{z:=z+1;y:=y*z\}.$$

$$[\![1=0!]\!]y:=1[\![y=0!]\!]$$

$$[\![y=0!]\!]z:=0[\![y=z!]\!]$$

$$[\![y=z!]\!]Q[\![y=z! \land z=x]\!]$$

$$[\![y=z!]\!]Q[\![y=z! \land z=x]\!]$$

 $[x > 0]y := 1; z := 0; Q[y = z! \land z = x]$

Drawbacks of Program Proof

- Proving that arbitrary program X has property Y is undecidable
- Precisely specifying all of program's intended actions is notoriously hard
 - Doing such a detailed spec & associated proofs usually much harder than writing & testing the program!
- Dynamic memory management (heap) is difficult to reason about
- Concurrency is even more difficult to reason about
 - See well-known books by Manna and Pnueli (1992,1995) or text by Apt et al (2009)

Research in Program Proof is Active

- Still too complicated for realistic programs/languages
- But it is growing fast!
 - The Atelier B system was used to develop part of the embedded software of the Paris metro line 14 and other railroad-related systems
 - Formally proved C compiler was developed using the Coq proof assistant (http://compcert.inria.fr)
 - Microsoft's hypervisor for highly secure virtualization was verified using VCC and the Z3 prover
 - ► L4-verified project developed a formally verified micro-kernel with high security guarantees, using analysis tools on top of the Isabelle/HOL proof assistant (https://sel4.systems)
 - ► https://deepspec.org/main
 - ► Facebook's Infer tool based on Separation Logic https://fbinfer.com

Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model

Assertions


Non-Progress Cycles

End States

Model-Checking: Stating Properties Outside the Mode

Model-Checking

- 1. Develop a model of the program
 - ► This helps abstract away from unnecessary details
 - Provides a different way of thinking about your problem
 - ► Must be careful to not oversimplify
- 2. Prove properties of the model
 - Use tools to analyze the model



Software Model Checking

Software model checking is the algorithmic analysis of programs to prove properties of their executions

- ▶ There is an extensive literature on this topic
- We only focus on one example (explicit state, automated, model-checking for temporal logic based on automata techniques)
- ➤ Survey: Ranjit Jhala, Rupak Majumdar: Software model checking. ACM Comput. Surv. 41(4): 21:1-21:54 (2009)


Model-Checking

Two well-known explicit-state model-checkers for concurrent/distributed computing

- Spin (we'll use this one)
 - Developed by Gerard Holzmann (1980s)
 - Awarded ACM's Software System Award in 2001
 - Example of use: Mars Code, Gerard J. Holzmann, Communications of the ACM, Vol. 57 No. 2, Pages 64-73, Feb 2014
- ► TLA+
 - ▶ Developed by Leslie Lamport (1994)
 - Example of use: How Amazon Web Services Uses Formal Methods, Chris Newcombe, Tim Rath, Fan Zhang, Bogdan Munteanu, Marc Brooker, Michael Deardeuff, Communications of the ACM, Vol. 58 No. 4, Pages 66-73, April 2015

Model-Checking: Plan

- 1. Introduction to Promela
- 2. Stating properties within the model
 - Assertion statements
 - Meta labels
 - end labels
 - progress labels
 - accept labels
- 3. Stating properties outside the model
 - Never claims
 - Temporal logic formulas



Bibliography - Spin and Promela

Tutorial on Promela and Spin:

► Model Checking Concurrent Programs, Ian Barland, Moshe Vardi and John Greiner. Available here: https://cnx.org (search for title above) or download it here https://cnx.org/exports/cd5745fd-3270-46ee-9b64-6e4843b67c43@3.4.pdf/model-checking-concurrent-programs-3.4.pdf

Principles of the Spin Model Checker, Mordechai Ben-Ari, Springer-Verlag London, 2008.



Some slides:

http://spinroot.com/spin/Doc/SpinTutorial.pdf

Bibliography - Foundations of Model Checking

 Principles of Model Checking, Christel Baier and Joost-Pieter Katoen, MIT Press, April 2008



Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model

Assertions

Non-Progress Cycles

End States

Model-Checking: Stating Properties Outside the Model

Promela

- Spin uses Promela (PROcess MEta LAnguage) for representing models
- The aim of Promela is to model concurrent and distributed systems
- We'll look at some examples of Promela code
- Spin can be used in two modes
 - Simulation mode: this runs the Promela model
 - ▶ Verification mode: this checks the Promela model

Promela Models

Consist of:

- type declarations
- channel declarations
- variable declarations
- process declarations
- ▶ init process

Corresponds to a (usually large, but) finite transition system, so

- no unbounded data
- no unbounded channels
- no unbounded processes
- no unbounded process creation

```
mtype = {MSG, ACK};
   chan toS = ...
   chan toR = ...
   bool flag;
   proctype Sender() {
       ... process body ...
8
9
   proctype Receiver() {
10
11
12
13
   init {
14
15
16
```

Simple Sequential Program (eg1.pml)

```
active proctype P() {
1
     byte N = 10;
2
    byte sum = 0;
    byte i;
     for (i : 1 .. N) {
5
       sum = sum + i;
6
     }
    printf("The sum of the first %d numbers = %d\n",
8
        N, sum);
9
10
```

- ▶ P is referred to as the process type
- active spawns a process type

Simple Sequential Program

```
active proctype P() {
     byte N = 10;
2
   byte sum = 0;
3
    byte i=1;
     dο
     :: i > N -> break
     :: else ->
7
              sum = sum + i;
8
9
             i++
    od;
10
    printf("The sum of the first %d numbers = %d\n",
11
        N, sum);
12
13
```

► Same as previous example only uses do-od

Simple Interleaving (eg2.pm1)

```
byte n = 0;
2
   active proctype P() {
4
       n = 1;
       printf("Process P, n = %d\n", n);
5
   }
6
7
   active proctype Q() {
       n = 2;
9
       printf("Process Q, n = \frac{d}{n}, n);
10
11
```

Simple Interleaving with Race Condition (eg3.pml)

```
byte n = 0;
1
2
   active proctype P() {
       byte temp;
4
       temp = n + 1;
5
       n = temp;
6
       printf("Process P, n = %d\n", n)
8
9
   active proctype Q() {
10
       byte temp;
11
       temp = n + 1;
12
       n = temp;
13
       printf("Process Q, n = %d n", n)
14
15
```

➤ Statements are atomic in Promela; interleaving occurs in an if- or do-statement (more later)

Simple Interleaving with Race Condition (eg4.pml)

- ► Same as previous example but shorter
- ► Note the use of [2] and _pid (predefined variables start with an underscore)

```
byte    n = 0;

active [2] proctype P() {
    byte temp;
    temp = n + 1;
    n = temp;
    printf("Process P%d, n = %d\n", _pid, n);
}
```

Simple Interleaving with Race Condition (eg5.pml)

- ▶ init is the first process that is activated
- ▶ run instantiates a process
- Convention: run expressions are enclosed in atomic so that all processes are instantiated before any of them begins execution

```
byte n;
2
   proctype P(byte id; byte incr) {
     byte temp;
    temp = n + incr;
     n = temp;
     printf("Process P%d, n = %d \ n", id, n)
8
9
   init {
10
   n = 1;
11
   atomic {
12
     run P(1, 10);
13
       run P(2, 15)
14
15
16
```

Simple Interleaving with Race Condition (eg6.pml)

- ► The body of a process consists of a sequence of statements. A statement is either
 - executable: the statement can be executed immediately.
 - blocked: the statement cannot be executed.
- An assignment is always executable.
- An expression is also a statement; it is executable if it evaluates to non-zero.
 - 2 < 3 always executable
 - x < 27 only executable if value of x is smaller 27
 - 3 + x executable if x is not equal to -3

Simple Interleaving with Race Condition (eg6.pml)

(_nr_pr == 1) causes init to block until the expression is true
(_nr_pr is number of processes currently running)

```
bvte
            n = 0:
   proctype P() {
3
        byte temp, i;
        for (i:1..10) {
5
          temp = n;
          n = temp + 1
7
      }
8
   init {
        atomic {
10
             run P();
11
             run P()
12
13
        (_nr_pr == 1);
14
        printf("The value is %d\n", n);
15
16
```

Model-Checking: Stating properties within the model

- Assertion statements
- ► Meta labels
 - end labels
 - progress labels
 - accept labels

Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model Assertions


Non-Progress Cycles End States

Model-Checking: Stating Properties Outside the Model

Assert (eg8.pml)

```
1 byte n = 0;
2 byte finished = 0;
3
  active [2] proctype P() {
    byte i = 1;
5
6
  byte temp;
7 for (i:1..10) {
    temp = n;
8
9
    n = temp + 1
    }
10
    finished++; /* Process terminates */
11
  }
12
13
   active proctype Finish() {
14
    finished == 2; /* Wait for termination */
15
    printf("n = %d\n", n);
16
assert (n > 2); /* Assert can't be 2 */
  }
18
```

Verification in Spin using Assertions



```
spin -a eg8.pml
sgcc -o pan pan.c
sgc./pan
```

Verification in Spin using Assertions

Since there was an assertion violation, it also generates a trail counterexample (eg8.pml.trail)

```
pan:1: assertion violated (n>2) (at depth 88)
    pan: wrote eg8.pml.trail
    (Spin Version 6.4.8 -- 2 March 2018)
    Warning: Search not completed
        + Partial Order Reduction
    Full statespace search for:
        never claim
                               - (none specified)
10
       assertion violations
11
                             - (not selected)
       acceptance cycles
12
       invalid end states +
13
    State-vector 36 byte (Size of a state), depth
    reached 92 (longest path), errors: 1
       138429 states (total number of states), stored
16
17
       87813 states, matched
       226242 transitions (= stored+matched)
18
19
           0 atomic steps
20
    hash conflicts:
                   4500 (resolved)
21
22
    Stats on memory usage (in Megabytes):
        8.449 equivalent memory usage for states (stored*(State-vector + overhead))
23
24
               actual memory usage for states (compression: 65.86%)
        5.565
25
               state-vector as stored = 14 byte + 28 byte overhead
26
      128.000 memory used for hash table (-w24)
27
       0.534 memory used for DFS stack (-m10000)
       134.003 total actual memory usage (memory used)
28
```

Inspecting the Trail

We can replay the counterexample in eg8.pml.trail

```
$ ./pan -r
1
                                              [i = 1]
     1:
           proc
                  1 (P) eg8.pml:7 (state 1)
1
2
     2:
           proc
                  1 (P)
                        eg8.pml:7 (state 8) [((i<=10))]
     3:
                  0 (P) eg8.pml:7 (state 1) [i = 1]
3
           proc
                  0 (P) eg8.pml:7 (state 8) [((i<=10))]
     4:
4
           proc
     5:
                  1 (P) eg8.pml:8 (state 3)
                                              [temp = n]
           proc
5
     6:
                  0 (P)
                        eg8.pml:8 (state 3)
                                              [temp = n]
           proc
6
     7:
                  1 (P) eg8.pml:9 (state 4)
                                              [n = (temp+1)]
           proc
7
     8:
                  1 (P) eg8.pml:7 (state 5) [i = (i+1)]
           proc
8
                  1 (P) eg8.pml:7 (state 8) [((i<=10))]
     9:
           proc
9
                    (P)
                        eg8.pml:8 (state 3)
                                              [temp = n]
    10:
           proc
10
11
    . . .
```

Critical Section - Revisiting Attempt III

```
boolean wantP = false;
  boolean wantQ = false;
2
  Thread start { //P 1
                           Thread.start { // Q
1
    while (true) {
                             while (true) {
2
   // non-critical section3 // non-critical section
3
   wantP = true;
                    4 wantQ = true;
4
   await (!wantQ);
                   5 await (!wantP);
   // CRITICAL SECTION 6 // CRITICAL SECTION
   wantP = false;
                           wantQ = false;
7
     // non-critical section8
                             // non-critical section
8
9
                         9
                           }
10
                         10
```

- Mutex: Yes (we'll prove this using spin; we introduce assertions using a ghost variable critical)
- ► Absence livelock: No
- ► Free from starvation: No

Attempt III in Promela

```
bool wantP = false; bool wantQ = false;
   byte critical = 0;
3
   active proctype P() {
       do ::
5
           wantP = true;
6
            do
           :: wantQ == false -> break
8
           :: else
           od;
10
           critical++;
11
           assert (critical == 1);
12
           critical --;
13
14
            wantP = false
       od
15
   }
16
17
   // Similar with Q
18
```

Critical Section

We verify:

```
$ spin -a eg9.pml
2 $ gcc -o pan pan.c
3 $ ./pan
```

Result:

Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model

Assertions

Non-Progress Cycles

End States

Model-Checking: Stating Properties Outside the Mode

Non-Progress Cycles

- ► Spin can check for some simple liveness properties without the need to use Linear Temporal Logic
- We designate states as progress states by using a label prefixed with progress
- An infinite computation that does not include infinitely many occurrences of a progress state is called a *non-progress cycle*

Revisiting Attempt III


```
1
  boolean wantP = false;
2
  boolean want0 = false:
  Thread start { //P 1
                           Thread.start { // Q
1
    while (true) {
                  while (true) {
2
   // non-critical sections // non-critical section
3
   wantP = true;
                         4 wantQ = true;
4
   await (!wantQ);
                   5 await (!wantP);
5
   // CRITICAL SECTION 6 // CRITICAL SECTION
   wantP = false;
7
                  7 wantQ = false;
                             // non-critical section
     // non-critical sections
8
9
                         9
                           }
10
                         10
```

- Mutex: Yes
- ► Absence livelock: No (we'll prove this using spin)
- ► Free from starvation: No

Attempt III in Promela

```
bool wantP = false; bool wantQ = false;
2
   active proctype P() {
       do ::
4
          wantP = true;
5
           do
6
           :: wantQ == false -> break
7
           :: else
8
           od;
9
   progress1:
10
           wantP = false
11
12
      od
13
   active proctype Q() {
14
       do ::
15
           wantQ = true;
16
            do
17
           :: wantP==false -> break
18
           :: else
19
           od;
20
   progress2:
21
            wantQ = false
22
                                                                 42 / 77
```

We Verify...



We Verify...

The trail

```
starting claim 2
spin: couldn't find claim 2 (ignored)
1 \ 0:1 \ 1) \ want 0 = 1
Process Statement
                       wantQ
0 P:1 1) wantP = 1 1
<<<<START OF CYCLE>>>>
Process Statement
                     wantP wantQ
1 Q:1 1) else
0 P:1 1) else
spin: trail ends after 8 steps
#processes: 2
 8: proc 1 (Q:1) a4.pml:19 (state 5)
 8: proc 0 (P:1) a4.pml:7 (state 5)
2 processes created
Exit-Status 0
```

Revisiting Attempt IV

```
boolean wantP = false;
boolean wantQ = false;
   Thread.start { //P
                              Thread.start {//Q
     while (true) {
                                while (true) {
2
     // non-critical section3
                                // non-critical section
3
     wantP = true:
                               wantQ = true:
4
     while (wantQ) {
                               while (wantP) {
5
      wantP = false;
                                   wantQ = false;
6
      wantP = true;
                                   wantQ = true:
7
8
      // CRITICAL SECTION 9
                                 // CRITICAL SECTION
     wantP = false;
                                 wantQ = false;
10
                    10
     // non-critical section
                                 // non-critical section
11
12
                            12
                              }
13
                            13
```

Mutex: Yes

► Absence of livelock: No

► Free from starvation: No

Revisiting Attempt IV

```
bool wantP = false, wantQ = false;
2
  active proctype P() {
3
    do
4
     :: wantP = true;
5
     do
6
    :: wantQ -> wantP = false; wantP = true
7
    :: else -> break
8
9
   od;
   wantP = false
10
     od
11
12 }
13
  active proctype Q() {
14
    do
15
   :: wantQ = true;
16
        do
17
   :: wantP -> wantQ = false; wantQ = true
18
        :: else -> break
19
        od;
20
   wantQ = false
21
     od
22
                                                        46 / 77
```

Revisiting Attempt IV

- ► Check that there is no deadlock using spin
- ► Check that absence of livelock fails

Starvation

- Livelock: computation continues but no process makes actual progress
- ► Starvation: computation continues, and some processes make progress but others don't.

- Mutex: Yes
- ▶ Absence livelock: Yes
- ► Free from starvation: No (a process could remain indefinitely in its non-critical section)

Starvation

- Mutex: Yes
- ▶ Absence livelock: Yes
- ► Free from starvation: No (a process could remain indefinitely in its non-critical section)

Show that Attempt I does not enjoy freedom from starvation Here is how you code a simple infinite loop:

```
1 do
2 :: else
3 od;
```

Exercise: Write Dekker's Algorithm in Promela

```
int turn = 1;
boolean wantP = false;
3 boolean wantQ = false;
  Thread.start { //P 1 Thread.start { //Q
1
    while (true) {
                         while (true) {
2
    // non-CS
                3 // non-CS
3
   wantP = true
                     4 wantQ = true
   while wantQ
                5 while wantP
5
     6
        wantP = false 7
7
                             wantQ = false
        await (turn==1)8
                             await (turn==2)
8
        wantP = true 9
9
                             wantQ = true
10
                    10
    // CS
                        // CS
                    11
11
    turn = 2
                   12 turn = 1
12
     wantP = false
                 13 wantQ = false
13
     // non-CS
                         // non-CS
                    14
14
15
                    15
16
                    16
```

Right to insist on entering is passed between the two processes

Exercise: Write Dekker's Algorithm in Promela

```
bool wantp = false, wantq = false;
byte turn = 1;
3
   active proctype P() {
       do
5
       :: wantp = true;
6
            do
7
            :: !wantq -> break;
8
            :: else ->
9
                if
10
                :: (turn == 1) /* no statements, leaves if */
11
                :: (turn == 2) ->
12
13
                    wantp = false;
                    do
14
                    :: turn==1 -> break
15
                    :: else
16
17
                    od:
                    wantp = true
18
                fi
19
           od;
20
           wantp = false;
21
            turn = 2
22
                                                               51 / 77
```

Program Correctness

Model-Checking

Introduction to Promela

Model-Checking: Stating properties within the model

Assertions

Non-Progress Cycles

End States

Model-Checking: Stating Properties Outside the Mode

Additional Comment on End States

```
byte request = 0;
2
   active proctype Server1() {
     do
4
      :: request == 1 ->
5
              printf("Service 1\n");
6
              request = 0;
7
8
     od
9
   active proctype Server2() {
10
     do
11
12
      :: request == 2 ->
              printf("Service 2\n");
13
              request = 0;
14
     od
15
16
   active proctype Client() {
17
     request = 1;
18
     request == 0;
19
     request = 2;
20
     request == 0;
21
22
                                                                  53 / 77
```

Additional Comments on End States

- A process that does not terminate in its last instruction is said to be in an invalid end state
- ➤ Servers are always blocked at the guard of the do-statement waiting for it to become executable
- ► To avoid this: use a label to indicate that a control point is a valid end point, even if it is not the last instruction

```
1 active proctype Server1() {
2    endserver:
3    do
4    :: request == 1 -> ...
5    od
6 }
```

Model Checking: Stating Properties Outside the Model

Stating properties inside the model has various drawbacks:

- misplaced
- updating is error prone
- no separation of concerns (model vs. property of the model)
- many interesting properties not expressible via assertions

Model Checking: Stating Properties Outside the Model

- never claims
- temporal logic formulas

Appendix

Installing Spin and jSpin

More Details on Promela Syntax

Installing Spin

- ▶ Binaries: https://github.com/nimble-code/Spin
- ► OS X: Make executable (chmod +x spin651_mac64)

Installing jSpin

```
http://www.weizmann.ac.il/sci-tea/benari/software-and-learning-materials/jspin
```

- Compile and create .jar file
- Configuration:
 - Create a jspin-5-0/bin directory
 - Add binary file for spin in jspin-5-0/bin (eg. spin651_mac64).
 - Modify the following items in config.cfg:

```
SPIN=../bin/spin651_mac64
C_COMPILER=/usr/bin/gcc
DOT=/usr/local/bin/dot
```

► Somewhat outdated reference manual: http: //wwinf.u-szeged.hu/~gombas/HSRV/jspin-user.pdf

Emacs and Dot

- Emacs
 - Promela mode:

```
https://github.com/rudi/promela-mode
```

- Place in ~/.emacs.d/plugins
- Install by adding this to .emacs
 (add-to-list 'load-path "~ /.emacs.d/plugins")
 (require 'promela-mode)
- ► Install dot
 - brew install graphviz

Execution using Spin/jSpin

- ► From command line: ../bin/spin651_mac64 count.pml
- Using jSpin
- Modes:
 - Random
 - Interactive
 - Guided: follows the error trail that was produced by an earlier verification (not presented yet) run

Appendix

Installing Spin and jSpin

More Details on Promela Syntax

Promela Summary

FEATURE	С	PROMELA
integers	char, short, int, long	byte, short, int
bit field	unsigned	unsigned
floats	float, double	NONE
boolean	int	bool
strings	char, char*	NONE
arrays	yes	1D & limited
operators	many	mostly same
if	as usual	similar to Erlang
loops	while, for, do	do, similar to if
output	printf	printf
input	scanf	NONE
functions	yes	NO
pointers	yes	NO
enum	enum	mtype
comments	/* */ and //	/* */
срр	full	1-line #define, #include

If Syntax

▶ "Guarded commands" wrapped inside "if ... fi"

```
1 disc = b*b - 4*a*c;
2 if
3 :: disc > 0 ->
4    printf("two real roots\n")
5 :: disc < 0 ->
6    printf("no real roots\n")
7 :: disc == 0 ->
8    printf("duplicate real roots\n")
9 fi
```

If Semantics

- ► First: evaluate all guards
- ► Then:
 - If no guard true: statement blocks until at least one guard becomes true (which could happen due to action of some concurrent process)
 - ► If one guard true: execute its command(s)
 - ► If more than one guard true: execute command(s) of randomly chosen guard

Else

- Guard consisting of "else" keyword is true if all other guards are blocked
- Example:

```
1 disc = b*b - 4*a*c;
2 if
3 :: else ->
4    printf("two real roots\n")
5 :: disc < 0 ->
6    printf("no real roots\n")
7 :: disc == 0 ->
8    printf("duplicate real roots\n")
9 fi
```

Do Syntax

- Similar to if statement
- Example: compute GCD by repeated subtraction

- ► Notes:
 - No loop test; only way out is via break
 - Body consists of guarded commands
 - Some true guard is chosen at random
 - Block if no true guard

Do Semantics

- Promela has no other type of loop
- Most common loop has only 2 guarded commands:

```
1 do
2 :: [exit test] -> break
3 :: else -> [body statements]
4 od
```

► This structure provides deterministic operation like:

```
while (not [exit test])
{ [body statements] }
```

Another Example

```
proctype P() {
   int x = 15, y = 20;
   int a = x, b = y;

do
   :: a > b -> a = a - b
   :: b > a -> b = b - a
   :: a == b -> break
   od
   printf("GCD(%d, %d) = %d\n", x, y, a);
}
```

Note:

- proctype P() declares no-argument program P
- Can include arguments:

```
proctype P(int x, int y) {
    int a = x, b = y;
    etc. }
```

Spawning a Process

- ► Can start processes using run operator: run P(15, 20)
- Also, can declare process with active proctype
 - Adding "active" means "define and run this program"
- ➤ To start two processes executing same code, use: active [2] proctype P(int x, int y)
- Can create an initial process that runs before any of the "proctype" processes
 - This process must be named init

Predefined Variables

- _pid is process ID
- ▶ _nr_pr is number of active processes
- Examples:

```
printf("process %d: n goes from %d to %d\n", _pid, temp, n)

if
:: _nr_pr == 1 -> printf("at end n = %d\n", n);

fi
```

Blocking Statements, I

- Concurrent programs must often wait for some event
- ► Possible to guard any statement
- ► This:

```
1 _nr_pr == 1 -> printf("at end n = %d\n", n)
```

is the same as:

```
if
:: _nr_pr == 1 -> printf("at end n = %d\n", n)
if
```

Blocking Statements, II

- "->" arrow is just syntactic sugar
- ➤ Can write expression by itself; if it doesn't evaluate to non-zero then program will block
- ► This:

```
1  _nr_pr == 1;
2  printf("at end n = %d\n", n)
```

is the same as:

```
_1 _nr_pr == 1 -> printf("at end n = %d\n", n)
```

Atomicity, I

- Individual Promela statements are atomic
- Warning! In Promela, expressions are statements too (hence expressions are atomic)
- Example here, division by zero is possible:

```
1 if
2    :: a != 0 -> c = b / a;
3    :: else    -> c = b
4 fi
```

► In an if (and do) statement, interleaving may occur between the evaluation of the guard and the execution of the statement after the guard

Atomicity, II

- It is tempting to regard the entirety of a != 0 -> c = b / a as atomic
- But it consists of two atomic parts, a != 0 and c = b / a
- Remember that this:

```
a != 0 -> c = b / a
```

could be written as:

```
1 a != 0; /* may block */
2 c = b / a
```

▶ The latter more obviously contains two atomic parts

Atomicity, III

► To group statements together atomically use atomic

```
1 atomic {
2    a != 0;    /* may block */
3    c = b / a
4 }
```

- ▶ If any statement within the atomic sequence blocks, atomicity is lost, and other processes may start executing statements.
- When the blocked statement becomes executable again, the execution of the atomic sequence can be resumed at any time (but it has to compete with other active processes)

Atomic & Run

- run only starts a concurrent process
- atomic prevents execution of any other actions besides those in its body
- ► Therefore, to start a group of processes that should run concurrently:

```
atomic {
    run P1(...);
    run P2(...);

    run PN(...)
}
```

► At conclusion of atomic block: all processes have been started but none is yet running

Variable Size

- ▶ Use smallest integer variable that will fit the need
- ► E.g., for integers known to be small use "byte" (8 bits) instead of "int" (32 bits)
- ▶ Reason: "verification" simulates all possible values of variable